A PERTURBED-ALEXANDER INVARIANT

DROR BAR-NATAN AND ROLAND VAN DER VEEN

Abstract. In this research announcement we give concise formulas, which lead to a simple and fast computer program, which computes a powerful knot invariant. This invariant ρ_1 is not new, yet our formulas are by far the simplest and fastest: given a knot K we write one of the standard matrices A whose determinant is the Alexander polynomial of K, yet instead of computing its determinant we consider a certain quadratic expression in the entries of A^{-1}. The proximity of our formulas to the Alexander polynomial suggest that they should have a topological explanation. This we don't have yet.

1. The Formulas

The selling point for this article is that the formulas in it are concise. Thus we start by running through these formulas for a knot invariant ρ_1 as quickly as we can. In Section 2 we turn the formulas into a short yet very fast computer program, and in Section 3 we quickly sketch the context: Alexander, Burau, Jones, Melvin, Morton, Rozansky, Overbay, Ohtsuki, and our own prior work.



Given an oriented n-crossing knot K, we draw it in the plane as a long knot diagram D in such a way that the two strands intersecting at each crossing are pointed up (that's always possible because we can always rotate crossings as needed), and so that at its beginning and at it's end the knot is oriented upward. An example is here on the right. We then label each edge of the resulting diagram with two integer labels: a running index k which runs from 1 to $2n+1$, and a "rotation number" r_k, which counts how many times the geometric rotation number of that edge (the signed number of times the tangent to the edge is horizontal and heading right, with cups counted with $+1$ signs and cups with -1; this number is well defined because at their ends, all edges are headed up). On the right the running index runs from 1 to 7, and the rotation numbers for all edges is 0 except for r_4, for which it is -1.

Next, we form a $(2n) \times (2n+1)$ matrix B of Laurent polynomials in a formal variable T as follows. Every crossing c of K has a sign $s \in \{\pm 1\}$ and is surrounded by four edges, with labels i and j below the crossing (where the label i belongs to the over strand and j to the under strand) and labels $k = i+1$ and $l = j+1$ above the crossing. Such a crossing defines rows i and j and columns i, j, k and l of B as below, with the rest

Date: First edition ?, this edition February 25, 2022.

2010 *Mathematics Subject Classification.* TBW.

Key words and phrases. TBW.

This work was partially supported by NSERC grant RGPIN-2018-04350. It is available in electronic form, along with source files and a verification *Mathematica* notebook at http://drorbn.net/APAI and at arXiv: 2201.?????.

of B set to be 0:



$\longrightarrow$

B	col i	col j	col k	col l
row i	1	0	-1	0
row j	0	1	T^s-1	$-T^s$

(1)

We then remove the first column of B and call the result A (so $B = (\phi|A)$, with ϕ a single column), we invert A and prepend to it a row of 0's, and we call the resulting $(2n+1)\times(2n)$ matrix $G = (g_{\alpha\beta})$, as on the right. In the case of the trefoil knot as shown above, the matrices B and G are

$$G = \begin{pmatrix} 0 & \cdots & 0 \\ & A^{-1} & \\ & & \end{pmatrix}$$

$$\begin{pmatrix} 1 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & -T & 0 & 0 & T-1 & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 & 0 \\ 0 & T-1 & 0 & 1 & -T & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & T-1 & 0 & 1 & -T \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & 0 & 0 & 0 & 0 & 0 \\ -1 & -\frac{T}{T^2-T+1} & \frac{1-T}{T^2-T+1} & \frac{1-T}{T^2-T+1} & -\frac{(T-1)T}{T^2-T+1} & 0 \\ -1 & -\frac{T}{T^2-T+1} & -\frac{T^2}{T^2-T+1} & \frac{1-T}{T^2-T+1} & -\frac{(T-1)T}{T^2-T+1} & 0 \\ -1 & \frac{1}{-T^2+T-1} & -\frac{T}{T^2-T+1} & -\frac{T}{T^2-T+1} & \frac{1-T}{T^2-T+1} & 0 \\ -1 & \frac{1}{-T^2+T-1} & -\frac{T}{T^2-T+1} & -\frac{T}{T^2-T+1} & -\frac{T^2}{T^2-T+1} & 0 \\ -1 & -\frac{1}{T} & -1 & -\frac{1}{T} & -1 & -\frac{1}{T} \end{pmatrix}.$$

We note without supplying details that the matrix B comes in a straightforward way from Fox calculus as it is applied to the Wirtinger presentation of the fundamental group of the complement of K (using the diagram D), and hence the determinant of A is equal up to a unit to the Alexander polynomial of K. In fact, we have that

$$\Delta = T^{(r(D)-w(D))/2}\det(A), \tag{2}$$

where $r(D) := \sum_k r_k$ is the total rotation number of D, where $w(D) = \sum_c s_c$ is the writhe of D, namely the sum of the signs s_c of all the crossings c in D, and where Δ is the normalized Alexander polynomial of K, so $\Delta(T) = \Delta(T^{-1})$ and $\Delta(1) = 1$. In our example $\det(A) = T^3 - T^2 + T$, $r(D) = -1$, and $w(D) = 3$, so $\Delta = T^{(-1-3)/2}(T^3 - T^2 + T) = T - 1 + T^{-1}$, as expected for the trefoil knot.

We can now define our invariant ρ_1. It is the sum of two sums. The first is over all crossings c in D, and for such a crossing we let s denote its sign and we let i and j denote the edge labels of the incoming over- and under-strands, respectively. The second is a sum over the edges k of D of a correction term dependent on the rotation number r_k. We multiply both of these summands by Δ^2 to "clear the denominators"[1]:

$$\begin{aligned} \rho_1 := \Delta^2 \sum_c s\left((1-T^s)\, g_{ij}\,(g_{ij}-g_{jj}) + 2g_{ii}g_{ij} - g_{ij}g_{ji} - g_{ii}g_{jj} - g_{ij} + g_{jj} - 1/2\right) \\ +\Delta^2 \sum_k r_k\,(g_{kk} - 1/2)\,. \end{aligned} \tag{3}$$

A direct calculation shows that in our example $\rho_1 = -T^2 + 2T - 2 + 2/T - 1/T^2$.

Theorem 1. *The quantity ρ_1 is a knot invariant.*

This is a research announcement, so we will not prove Theorem 1 here. We merely comment that ρ_1 has more separation power than the Jones polynomial, yet it is closer to the more topologically-meaningful Alexander polynomial Δ: it is cooked up from the same matrix A

[1]The first summand is quadratic in the entries of G and hence it has denominators proportional to Δ^2.

and in terms of computational complexity, computing ρ_1 is not very different from computing Δ. In order to compute Δ we need to compute the determinant of A, while to compute ρ_1 we need to invert A and then compute a sum of $O(n)$ terms that are quadratic in the entries of A^{-1}. We have computed ρ_1 for knots with over 200 crossings using the unsophisticated implementation presented in Section 2.

Topologists should be intrigued! ρ_1 is cooked from the same matrix as the Alexander polynomial Δ, yet we have no topological interpretation for ρ_1.

2. Implementation

Two of the main reasons we like ρ_1 is that it is very easy to implement and even an unsophisticated implementation runs very fast. To highlight these points we include a full implementation here, a step-by-step run-through, and a demo run. We write in Mathematica [Wo], and you can find the notebook displayed here at [BDV, APAI.nb].

We start by loading the library `KnotTheory‘` [BM] (it is used here only for the list of knots that it contains, and to compute other invariants for comparisons). We also load minor conversion routine [BDV, RVK.nb / RVK.m] whose internal workings are simple and yet irrelevant here.

```
Once[<< KnotTheory`; << RVK.m];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

2.1. **The Program.** This done, here is the full ρ_1 program:

```
ρ[K_] := Module[{Cs, r, n, B, A, c, s, i, j, Δ, G, g, ρ1},
   {Cs, r} = List @@ RVK[K]; n = Length[Cs]; B = Table[0, 2 n, 2 n + 1];
   Do[{s, i, j} = c;
    B⟦{i, j}, {i, j, i + 1, j + 1}⟧ = {{1, 0, -1, 0}, {0, 1, T^s - 1, -T^s}}, {c, Cs}];
   A = B⟦All, 2 ;;⟧;
   Δ = T^((Total[r]-Total[First/@Cs])/2) Det[A];
   G = Prepend[Table[0, 2 n]][Inverse[A]]; g_{α_,β_} := G⟦α, β⟧;
   ρ1 = Δ^2 Sum[{s, i, j} = c;
      s ((1 - T^s) g_ij (g_ij - g_jj) + 2 g_ii g_ij - g_ij g_ji - g_ii g_jj - g_ij + g_jj - 1/2), {c, Cs}];
   ρ1 += Δ^2 Sum[r⟦k⟧ (g_kk - 1/2), {k, 2 n}] ;
   Factor@{Δ, ρ1}];
```

The program uses mostly the same symbols as the text, so even without any knowledge of Mathematica, the reader should be able to recognize at least formulas (1), (2), and (3) within it. As a further hint we add that the variables `Cs` ends up storing the list of crossing in a knot K, where each crossing is stored as a triple $\{s, i, j\}$, where s, i, and j have the same meaning as in (1). The conversion routine `RVK` automatically produces `Cs`, as well as a list `r` of rotation numbers, given any other knot presentation known to the package `KnotTheory‘`.

Note that the program outputs the ordered pair (Δ, ρ_1). The Alexander polynomial Δ is anyway computed internally, and we consider the aggregate (Δ, ρ_1) as more interesting than any of its pieces by itself.

2.2. **A Step-by-Step Run-Through.** We start by setting K to be the knot diagram on page 1 using the PD notation of `KnotTheory‘`. We then print `RVK[K]`, which is a list of crossings followed by a list of rotation numbers:

```
K = PD[X[4, 2, 5, 1], X[2, 6, 3, 5], X[6, 4, 7, 3]];
RVK[K]
```

```
RVK[{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}]
```

Next we set `Cs` and `r` to be the list of crossings, and the list of rotation numbers, respectively.

```
{Cs, r} = List @@ RVK[K]
```

```
{{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}}
```

We set `n` to be the number of crossings, `B` to be the zero matrix of dimensions $2n\times(2n+1)$, and then we iterate over `c` in `Cs`, adding a block as in (1) for each crossings.

```
n = Length[Cs]; B = Table[0, 2 n, 2 n + 1];
Do[{s, i, j} = c;
  B[[{i, j}, {i, j, i + 1, j + 1}]] = (1 0 -1       0
                                       0 1 T^s - 1  -T^s), {c, Cs}];
```

Here’s what `B` comes out to be:

```
B // MatrixForm
```

$$\begin{pmatrix} 1 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & -T & 0 & 0 & -1+T & 0 \\ 0 & 0 & 1 & -1 & 0 & 0 & 0 \\ 0 & -1+T & 0 & 1 & -T & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & -1+T & 0 & 1 & -T \end{pmatrix}$$

Next we set `A` to be the matrix whose rows are `All` the rows of `B`, and whose columns are the columns of `B` starting from column 2. We set Δ to be the determinant of `A`, with a correction as in (2). So Δ is the Alexander polynomial of K.

```
A = B[[All, 2 ;;]];
Δ = T^((Total[r]-Total[First/@Cs])/2) Det[A]
```

$$\frac{T - T^2 + T^3}{T^2}$$

`G` is now the `Inverse` of `A` with a row of 0’s added at the start. We set $\mathtt{g}_{\alpha\beta}$ to be the matrix entries of `G`, and print `G`:

```
G = Prepend[Table[0, 2 n]][Inverse[A]]; gα_,β_ := G⟦α, β⟧;
G // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ \frac{-T+T^2-T^3}{T-T^2+T^3} & 0 & 0 & 0 & 0 & 0 \\ \frac{-T+T^2-T^3}{T-T^2+T^3} & -\frac{T^2}{T-T^2+T^3} & \frac{T-T^2}{T-T^2+T^3} & \frac{T-T^2}{T-T^2+T^3} & \frac{T^2-T^3}{T-T^2+T^3} & 0 \\ \frac{-T+T^2-T^3}{T-T^2+T^3} & -\frac{T^2}{T-T^2+T^3} & -\frac{T^3}{T-T^2+T^3} & \frac{T-T^2}{T-T^2+T^3} & \frac{T^2-T^3}{T-T^2+T^3} & 0 \\ \frac{-T+T^2-T^3}{T-T^2+T^3} & -\frac{T}{T-T^2+T^3} & -\frac{T^2}{T-T^2+T^3} & -\frac{T^2}{T-T^2+T^3} & \frac{T-T^2}{T-T^2+T^3} & 0 \\ \frac{-T+T^2-T^3}{T-T^2+T^3} & -\frac{T}{T-T^2+T^3} & -\frac{T^2}{T-T^2+T^3} & -\frac{T^2}{T-T^2+T^3} & -\frac{T^3}{T-T^2+T^3} & 0 \\ \frac{-T+T^2-T^3}{T-T^2+T^3} & \frac{-1+T-T^2}{T-T^2+T^3} & \frac{-T+T^2-T^3}{T-T^2+T^3} & \frac{-1+T-T^2}{T-T^2+T^3} & \frac{-T+T^2-T^3}{T-T^2+T^3} & \frac{-1+T-T^2}{T-T^2+T^3} \end{pmatrix}$$

It remains to blindly follow the two parts of Equation (3):

```
ρ1 = Δ² Sum[{s, i, j} = c;
    s ((1 - T^s) gij (gij - gjj) + 2 gii gij - gij gji - gii gjj - gij + gjj - 1/2), {c, Cs}]
```

$$\frac{\left(T-T^2+T^3\right)^2\left(-\frac{3}{2}+\frac{(1-T)\,T^2}{\left(T-T^2+T^3\right)^2}-\frac{2\,T\left(T-T^2\right)}{\left(T-T^2+T^3\right)^2}+\frac{T}{T-T^2+T^3}+\frac{T-T^2}{T-T^2+T^3}\right)}{T^4}$$

```
ρ1 += Δ² Sum[r⟦k⟧ (gkk - 1/2), {k, 2 n}]
```

$$\frac{\left(T-T^2+T^3\right)^2\left(\frac{1}{2}-\frac{T-T^2}{T-T^2+T^3}\right)}{T^4}+\frac{\left(T-T^2+T^3\right)^2\left(-\frac{3}{2}+\frac{(1-T)\,T^2}{\left(T-T^2+T^3\right)^2}-\frac{2\,T\left(T-T^2\right)}{\left(T-T^2+T^3\right)^2}+\frac{T}{T-T^2+T^3}+\frac{T-T^2}{T-T^2+T^3}\right)}{T^4}$$

And to output both Δ and ρ_1. We factor them just to put them in a nicer form:

```
Factor@{Δ, ρ1}
```

$$\left\{\frac{1-T+T^2}{T},\ -\frac{(-1+T)^2\left(1+T^2\right)}{T^2}\right\}$$

2.3. A Demo Run.

Here are Δ and ρ_1 of all the knots with up to 6 crossings:

```
Do[Echo[K → ρ[K]], {K, AllKnots[{3, 6}]}]
```

KnotTheory: Loading precomputed data in PD4Knots`.

$$\text{Knot}[3,1]\to\left\{\frac{1-T+T^2}{T},\ \frac{(-1+T)^2\left(1+T^2\right)}{T^2}\right\}$$

$$\text{Knot}[4,1]\to\left\{-\frac{1-3\,T+T^2}{T},\ 0\right\}$$

$$\text{Knot}[5,1]\to\left\{\frac{1-T+T^2-T^3+T^4}{T^2},\ \frac{(-1+T)^2\left(1+T^2\right)\left(2+T^2+2\,T^4\right)}{T^4}\right\}$$



FIGURE 2.1. A 48-crossing knot from [GST].

$\texttt{Knot[5, 2]} \to \left\{\frac{2-3T+2T^2}{T}, \frac{(-1+T)^2\left(5-4T+5T^2\right)}{T^2}\right\}$

$\texttt{Knot[6, 1]} \to \left\{-\frac{(-2+T)(-1+2T)}{T}, \frac{(-1+T)^2\left(1-4T+T^2\right)}{T^2}\right\}$

$\texttt{Knot[6, 2]} \to \left\{-\frac{1-3T+3T^2-3T^3+T^4}{T^2}, \frac{(-1+T)^2\left(1-4T+4T^2-4T^3+4T^4-4T^5+T^6\right)}{T^4}\right\}$

$\texttt{Knot[6, 3]} \to \left\{\frac{1-3T+5T^2-3T^3+T^4}{T^2}, 0\right\}$

Next is one of our favourites, a knot from [GST] (see Figure 2.1), which is a potential counterexample to the ribbon=slice conjecture. It takes about one minute to compute ρ_1 for this 48 crossing knot (note that Mathematica prints `Timing` information is seconds):

$\texttt{Timing@}\rho[\texttt{EPD}[X_{14,1}, \overline{X}_{2,29}, X_{3,40}, X_{43,4}, \overline{X}_{26,5}, X_{6,95}, X_{96,7}, X_{13,8}, \overline{X}_{9,28}, X_{10,41}, X_{42,11}, \overline{X}_{27,12}, X_{30,15}, \overline{X}_{16,61}, \overline{X}_{17,72}, \overline{X}_{18,83}, X_{19,34}, \overline{X}_{89,20}, \overline{X}_{21,92}, \overline{X}_{79,22}, \overline{X}_{68,23}, \overline{X}_{57,24}, \overline{X}_{25,56}, X_{62,31}, X_{73,32}, X_{84,33}, \overline{X}_{50,35}, X_{36,81}, X_{37,70}, X_{38,59}, \overline{X}_{39,54}, X_{44,55}, X_{58,45}, X_{69,46}, X_{80,47}, X_{48,91}, X_{90,49}, X_{51,82}, X_{52,71}, X_{53,60}, \overline{X}_{63,74}, \overline{X}_{64,85}, \overline{X}_{76,65}, \overline{X}_{87,66}, \overline{X}_{67,94}, \overline{X}_{75,86}, \overline{X}_{88,77}, \overline{X}_{78,93}]]$

$$\Big\{60.7188,\ \Big\{-\frac{\left(-1+2T-T^2-T^3+2T^4-T^5+T^8\right)\left(-1+T^3-2T^4+T^5+T^6-2T^7+T^8\right)}{T^8},$$
$$\frac{1}{T^{16}}(-1+T)^2\big(5-18T+33T^2-32T^3+2T^4+42T^5-62T^6-8T^7+166T^8-242T^9+108T^{10}+$$
$$132T^{11}-226T^{12}+148T^{13}-11T^{14}-36T^{15}-11T^{16}+148T^{17}-226T^{18}+132T^{19}+108T^{20}-$$
$$242T^{21}+166T^{22}-8T^{23}-62T^{24}+42T^{25}+2T^{26}-32T^{27}+33T^{28}-18T^{29}+5T^{30}\big)\Big\}\Big\}$$

MORE.

3. Some Context

TBW.

Acknowledgement. We wish to thank TBW comments and suggestions.

References

[BDV] D. Bar-Natan and R. van der Veen, *A Perturbed-Alexander Invariant,* (self-reference), paper and related files at `http://drorbn.net/APAI`. The arXiv:????.????? edition may be older. See pp. 3.

[BM] D. Bar-Natan, S. Morrison, and others, `KnotThery‘`, a knot theory mathematica package, `http://katlas.org/wiki/The_Mathematica_Package_KnotTheory`. See pp. 3.

[GST] R. E. Gompf, M. Scharlemann, and A. Thompson, *Fibered Knots and Potential Counterexamples to the Property 2R and Slice-Ribbon Conjectures,* Geom. and Top. **14** (2010) 2305–2347, arXiv:1103.1601. See pp. 6.

[Wo] *Wolfram Language & System Documentation Center,* ωεβ/Wolf. See pp. 3.

Department of Mathematics, University of Toronto, Toronto Ontario M5S 2E4, Canada
Email address: `drorbn@math.toronto.edu`
URL: `http://www.math.toronto.edu/drorbn`

University of Groningen, Bernoulli Institute, P.O. Box 407, 9700 AK Groningen, The Netherlands
Email address: `roland.mathematics@gmail.com`
URL: `http://www.rolandvdv.nl/`